

## An alternate log axis using SAS<sup>®</sup> PROC GPLOT

Lucius Reinbolt, DataCeutics, Inc. (Formerly of InVentiv Health Clinical LLC), Nebraska

### ABSTRACT

It is typical in the pharmaceutical industry to plot Pharmacokinetic data using a log axis. Pharmacokinetic data is assumed to be log normally distributed. Setting the axis statement to use a log base of 10 allows SAS PROC GPLOT to create semi-log plots. These plots use intervals that correspond to the orders of magnitude as opposed to the standard linear scale. Each unit increase on the log scale represents an exponential increase in quantity for the given base. This gives the graph an appearance of having uneven tick marks. Also, these log plots will often force the upper or lower axis value to go up to the next power of 10, potentially leaving a lot of dead space in the graph. This can cause the range of values to look condensed and make it difficult to see a pattern. Sometimes it is necessary to remove this dead space while maintaining the log appearance of the graph. A solution to this is to take the log of the data and plot it as you would regular continuous data then customize the axis to show the tick marks as a log scale would present them. This gives the programmer more control over the range of values plotted while maintaining the characteristics of a log scale.

### INTRODUCTION AND BACKGROUND

Pharmacokinetic (PK) data is data collected to measure concentrations of the drug in the blood. This type of data is analyzed by scientists and assumed to be log normally distributed. Along with the analysis, it is helpful to view the data using a log scale. This is a simple request and can be done using the GLOT procedure along with specific options in the axis statement. These options will take the non-transformed data and have the graphing procedure create a log axis where SAS does all the work. However, the scale created may not be optimal for the data presented. Due to the nature of the scale, extra area may be plotted with no formal plotting points, which creates the illusion of blank space. This can typically be ignored; however, the request may be made to limit the scale to the actual data region. This paper focuses on a method that can simulate the log scale and also allows the programmer to have more control over the lower and upper limits used to plot.

The key to the log scale is in the distance that shows between the minor tick marks. The interval will appear uneven as compared to the even spacing on a normal linear scale. Each unit increase in the interval is shown as an exponential increase for a given base. The log scale itself will need to maintain this appearance to be called a log axis.

A set of sample data has been plotted to help illustrate this scale. Figure 1 shows a plot of some fake sample PK data using a normal axis.

Figure 2 shows the corresponding data plotted using the log scale options having the graphing procedure choose the scale by itself. The log scale, as automatically produced by SAS, will start and stop at the powers of 10 for a log base of 10. The lowest value is 5 so the graph will start at 1 and the largest value is 50 so the axis will go to 100.

This may result in the appearance of unused or blank space in a graph. While this blank space may not be considered severe and can normally be ignored, it may be requested that the scale be altered to remove this blank space.

A solution to this problem is defined below:

- Take the regular data and apply a log transformation.
- Plot this log transformed data in place of the regular data.
- Create a custom scale that shows the back transformed values as opposed to the log transformed values that are used.

This method requires the suppression of the axis as automatically produced and the insertion of a custom axis created using the annotate facility, which will allow greater control over the range of values used as compared to the standard axis. The annotate facility allows the programmer to append to the graph with many features that are not normally produced.




Figure 1. Example figure using a normal scale.




Figure 2. Example figure using a log scale.

## METHODS AND RESULTS

Most of the time, graphing using SAS procedures is simple and straight forward. On occasion, a request will alter a simple graph into a complex graph. The change may be to create a custom legend or in this case to create a custom axis. In these cases, the programmer may choose to augment the graph using the annotate facility. A common graph produced using pharmacokinetic (PK) data is a semi-log plot. This graph is typically done using the regular data and letting SAS automatically transform the data and create the appropriate scale. This scale is a very specific one in that the interval tick marks are related to the orders of magnitude as opposed to the standard linear scale. The standard linear scale has intervals between tick marks that increase in the same amount each time while the corresponding log scale equivalent has an uneven appearance. This is shown with figure 3 in a side by side comparison with a scale from 1 to 10 in this example graph.

The distance between the 1 and 2 value in the y-axis appears larger than the distance between 2 and 3 while in the normal axis this distance is the same.

Using a log base of 10, SAS will start and end the graph axis at a power of 10. In the case of this example data, the lower end of the interval is 1 and the upper end is 100. The sample data can be found in the appendix of this paper. The lowest non-zero value in the data is 5 and the largest value is 50. It may be desirable to adjust this scale and remove any dead space below the lowest value and/or above the highest value. This is not always the case, but, may be requested to help with the appearance graph and make certain patterns more easily visible to the scientists. Unfortunately, there is not a simple way to have the log axis range condense from 1 to 100 and go from 5 to 50 and keep the appearance of the uneven log scale.

SAS plots using a log axis do not require the programmer to transform the data, only to set options in the axis statement. But, if the programmer does transform the data and creates a custom axis, it is possible to condense the log axis. The log transformation is simply setting a new variable equal to the log of the concentration variable or another variable of interest.

```
data concentration; set concentration; lny=log(y); run;
```

The log transformed values will not have any immediate interpretable value to the scientist, so it is important that an axis be created to represent the back transformed scale. An example of this is the value 5 has a log transformed value of approximately 1.609. While the value 1.609 will be the y-axis value plotted, the value 5 will need to be placed on the y-axis scale to allow the viewer to see the data in its original scale. The first step in this is to remove the automatic tick marks and values in an axis. The programmer needs to set minor, major, and value options to 'none' in the axis statement. One also needs to remember the values that are being plotted and assign the order statement appropriately.

```
axis2 color=black
      label= (angle=90 h=3.5 color=black 'Concentration (unit)' j=c ' ' )
      minor=none
      major=none
      value=none
      order=(1.6 to 4.1 by .1);
```

A macro for replacing the tick marks on the x axis has been shown to work in the paper 'Improving Your Graphics Using SAS/Graph Annotate Facility'. Modifications have been made to create the y-axis as a log axis. Some modification is needed and the macro now looks like.

```
%annomac;
%macro tick(lval,val);
xsys='2'; ysys='2'; %move(0,&lval.);
xsys='B'; ysys='B'; %draw(-1.5,0,black,1,.25);
%cntl2txt; %label(-1.5,0,"&val.",black,0,0,2.15,arial,4);
%mend tick;
data anno;
%declanno;hsys='3';
%tick(4.094,60);
%tick(3.912,); **50;
run;
```

The macro will move and draw and label the actual tick marks at the appropriate values needed to show the back

transformed values. The 'lval' macro variable in the macro call above represents the log transformed value and the 'val' macro variable is the back transformed value. Simply omitting the 'val' value will result in a blank space being used instead of the value.

Now that the annotate data set is constructed, simply insert the data into the plotting procedure as shown in the programming example below.

```
title1 'Condensed Log Axis';  
proc gplot data= concentration ;  
  plot lny*time / haxis=axis1 vaxis=axis2 annotate=anno hzero;  
run; quit;
```

The resulting output is shown in figure 4. The resultant figure has a condensed log scale from 5 to 50. Any blank space has been removed while keeping the log scale appearance.




Figure 3. Example figure comparing a normal and a log scale.




Figure 4. Example figure using a custom generated log scale.

## CONCLUSION

Producing a graph in SAS can be challenging when significant alterations are requested from what the normal output should be. The annotate facility can be utilized to provide the programmer with options that allow for almost any solution needed to alter the normal graph. In this paper, a way to simulate the characteristics of the log axis and control the axis range further than the standard options has been shown. Annotation is a powerful tool that can really help the programmer to produce graphics that scientists can appreciate.

## REFERENCES

Mink, David and Pasta, David. 2006. Improving Your Graphics Using SAS/Graph® Annotate Facility. Paper 085-31 SUGI 31.

## ACKNOWLEDGMENTS

I would like to thank my colleagues in the statistical programming group at PharmaNet/i3 who provided comments while drafting this paper, especially Nancy Brucken. Your insights and guidance are greatly appreciated.

## CONTACT INFORMATION <HEADING 1>

Your comments and questions are valued and encouraged. Contact the author at:

Name: Lucius Reinbolt  
Enterprise: DataCeutics, Inc.  
Address: 1610 Medical Dr, Ste 300  
City, State ZIP: Pottstown, PA 19464  
Work Phone: 610 970 2333 Ext 6893  
E-mail: [reinboltl@dataceutics.com](mailto:reinboltl@dataceutics.com)  
Web: [www.dataceutics.com](http://www.dataceutics.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

## APPENDIX

```

options reset=all noborder cback=white gunit=pct htitle=3;

/* Create the sample data set */
data concentration;
  input time y;
  datalines;
0 0
0.5 5
1 15
1.5 25
2 36
2.5 42
3 47
3.5 49
4 50
4.5 49
5 47
6 40
8 25
10 13
12 7
16 0
;
run;

data concentration;
  set concentration;
  lny=log(y);
run;

%annomac;

%macro tick(lval,val);
xsys='2'; ysys='2'; %move(0,&lval.);
xsys='B'; ysys='B'; %draw(-1.5,0,black,1,.25);
%cntl2txt; %label(-1.5,0,"&val.",black,0,0,2.15,arial,4);
%mend tick;

data anno;
%dclanno;
hsys='3';
%tick(4.094,60);
%tick(3.912, ); **50;
%tick(3.689, );**40;
%tick(3.401, );**30;
%tick(2.996, );**20;
%tick(2.303,10);

%tick(1.609,5); **5;
%tick(1.792, ); **6;
%tick(1.946, ); **7;
%tick(2.079, ); **8;
%tick(2.197, ); **9;
run;

/* Define symbol characteristics */
symbol1 interpol=join v=circle color=black;

/* Define horizontal axis characteristics */
axis1 label=(h=3.5 'Time (hr)') minor=(number=3)
order=(0 to 16 by 4) offset=(0,0);

```

```
/* Define vertical axis characteristics */
axis2 color=black
  label= (angle=90 h=3.5 color=black 'Concentration (unit)' j=c ' ' )
  minor=none major=none value=none
  order=(1.6 to 4.1 by .1);

axis3 logbase=10 logstyle=expand color=black
  label= (angle=90 h=3.5 color=black 'Concentration (unit)' j=c ' ' )
  order=(1 10 100);

axis4 color=black
  label= (angle=90 h=3.5 color=black 'Concentration (unit)' j=c ' ' )
  order=(0 to 100 by 10);

/* Generate plot */
title1 'Condensed Log Axis';
proc gplot data= concentration ;
  plot lny*time / haxis=axis1 vaxis=axis2 annotate=anno hzero;
run; quit;

title1 'SAS Generated Log Axis';
proc gplot data= concentration;
  plot y*time / haxis=axis1 vaxis=axis3 ;
run; quit;

title1 'Normal Axis';
proc gplot data= concentration;
  plot y*time / haxis=axis1 vaxis=axis4 ;
run; quit;
```